
Послідовність Кеане
Умова
Нескінченна послідовність з нулів та одиниць, запропонована Кеане, виглядає так: 001001110001001110110110001… і формується за таким алгоритмом: спочатку записується 0, потім до нього приписується така ж послідовність (0) та її інверсія (1). Процес повторюється нескінченну кількість разів, кожен раз для нової отриманої послідовності.
Задачею є за номером цифри цієї послідовності (n≤10200) визначити, 0 це чи 1.
	Приклад

	INPUT.TXT
	18

	OUTPUT.TXT
	0


Розв’язок

1. Для часткового розв’язання задачі можна генерувати послідовність Кеане в цілочисельному масиві. Тоді після генерації достатньо у вихідний файл записати відповідний елемент масиву.
Масив у мові програмування Паскаль може містити не більше 65535 елементів, отже цей варіант розв’язку спрацює лише для невеликої частини тестів.
program Keane_1;
const maxN = 65000;

var A:array [1..maxN] of byte;

    i,j,N:word;

    fi,fo:text;

begin
  Assign (fi,’input.txt’); Reset (fi);

  ReadLn (fi,N);
  Close (fi);
  i:=1;
  while i<=N do
     begin
       for j:=1 to i do A[j+i]:=A[j];
       for j:=1 to i do A[j+2*i]:=1-A[j];

       i:=3*i;

     end;

  Assign (fo,’output.txt’); Rewrite (fo);

  WriteLn (fo,A[N]);

  Close (fo);

end.

2. Вочевидь, для повного розв’язку задачі необхідно певним чином аналізувати вхідний параметр – число n.
Проаналізуємо для прикладу послідовність Кеане після третього кроку генерації:

001001110001001110110110001
Умовно її можна поділити на три частини:
Якщо елемент з номером n міститься в першій частині, то очевидно, що для розв’язку достатньо розглянути кількість елементів втричі меншу за ту, яку ми зараз розглядаємо – 9 елементів.
Якщо елемент з номером n міститься у другій частині, то шукане значення дорівнюватиме значенню, яке стоїть на 9 позицій лівіше (відповідний елемент першої частини).
Якщо ж елемент з номером n міститься у третій частині, то його значення не дорівнюватиме відповідному значенню з першої частини (елемент з номером n-2·9), тобто матиме значення 1, якщо відповідний елемент має значення 0 і навпаки, 0, коли відповідний елемент має значення 1.
Таким чином, ми звели задачу до неї самої, але для меншого значення n. Такий спосіб розв’язування задач називається динамічним програмуванням. Як Ви побачите пізніше, він використовується при розв’язуванні найрізноманітніших класів задач.
Для відповіді на поставлене в умові запитання достатньо підрахувати кількість інверсій, які "пройшов" 0 перш ніж став на n-те місце. Через парну кількість інверсій 0 переходить у 0, а через непарну – в 1.
program Keane_2;

var i,j,inv,N:longint;

    fi,fo:text;

begin
  Assign (fi,’input.txt’); Reset (fi);

  ReadLn (fi,N);

  Close (fi);

  i:=1;
  while i<N do i:=i*3;

  inv:=0;
  while N>1 do
    begin

      i:=i div 3;

      if (i<N)and(N<=i*2) then N:=N-i else
      if i*2<N then begin Inc (inv); N:=N-2*i; end;
    end;
  Assign (fo,’output.txt’); Rewrite (fo);

  WriteLn (fo,inv mod 2);

  Close (fo);

end.

Поданий варіант рішення розширив діапазон значень n до діапазону longint (до 2’147’483’647), проте не дав повного розв’язку. З урахуванням обмежень на n, необхідно реалізувати довгу арифметику: множення на цифру та віднімання.
Виходячи з попередньої програми, можна допустити деяку оптимізацію. Так, наприклад, ми будемо запам’ятовувати проміжні степені трійки у масиві довгих чисел, щоб не писати ділення довгих чисел на 3.
Додатково нам знадобляться підпрограми:

· введення довгого числа (ReadLong);
· порівняння (Less).
Усі ці підпрограми (введення, порівняння, віднімання та множення на цифру) розглядалися раніше, тому на їх поясненні увага не загострюється.

program Keane_3;

const max3 = 300; maxL = 20;

type long = record
       N:array [1..maxL] of shortint;

       L:byte;

end;

procedure ReadLong (var f:text;var X:long);

var c:char; i:byte;

begin
  FillChar (X,SizeOf(X),0);

  while not EOLn(f) do
    begin
      Read (f,c);

      Inc (X.L);

      X.N[X.L]:=ord(c)-48;

    end;

  for i:=X.L downto 1 do X.N[maxL-X.L+i]:=X.N[i];

  for i:=1 to maxL-X.L do X.N[i]:=0;

end;

function Less (A,B:long):boolean;

var i:byte;

begin
  if A.L<>B.L then Less:=A.L<B.L else
     begin
       i:=maxL-A.L+1;

       while (i<=maxL)and(A.N[i]=B.N[i]) do Inc (i);

       Less:=(i<=maxL)and(A.N[i]<B.N[i]);

     end;

end;

procedure Multiply (A:long;B:byte;var C:long);

var i,p:byte;

begin
  FillChar (C,SizeOf(C),0);

  p:=0;

  for i:=maxL downto maxL-A.L do
      begin
        C.N[i]:=(A.N[i]*B+p) mod 10;

        p:=(A.N[i]*B+p) div 10;

      end;

  i:=1;

  while (i<=maxL)and(C.N[i]=0) do Inc (i);

  C.L:=maxL+1-i;

  if C.L=0 then C.L:=1;

end;

procedure Sub (A,B:long;var C:long);

var i,p:byte;

    t:shortint;

begin
  FillChar (C,SizeOf(C),0);

  p:=0;

  for i:=maxL downto maxL-A.L do
      begin
        C.N[i]:=A.N[i]-B.N[i]-p;

        if C.N[i]<0 then
           begin
             Inc (C.N[i],10);

             p:=1;

           end else p:=0;

      end;

  i:=1;

  while (i<=maxL)and(C.N[i]=0) do Inc (i);

  C.L:=maxL+1-i;

  if C.L=0 then C.L:=1;

end;

var j,inv:longint;

    i:array [0..max3] of long;

    ix2,N:long;

    fi,fo:text;

begin
  Assign (fi,'input.txt'); Reset (fi);

  ReadLong (fi,N);

  Close (fi);

  i[0].N[maxL]:=1; i[0].L:=1; j:=0;

  while Less (i[j],N) do
    begin
      Multiply (i[j],3,i[j+1]);

      Inc (j);

    end;

  inv:=0;

  while Less (i[0],N) do
    begin
      Dec (j);

      Multiply (i[j],2,ix2);

      if Less(i[j],N) and not Less(ix2,N)

         then Sub (N,i[j],N) else
      if Less(ix2,N) then
         begin
           Inc (inv);

           Sub (N,ix2,N);

         end;

    end;

  Assign (fo,'output.txt'); Rewrite (fo);

  WriteLn (fo,inv mod 2);

  Close (fo);

end.
Метод динамічного програмування застосовується у багатьох класах задач з програмування. Він дозволяє без зайвих витрат пам’яті і часу отримати відповідь через проміжні результати і стане у нагоді у найрізноманітніших випадках.
Арифметична прогресія
Умова
Дано послідовність натуральних чисел кількістю не більше 100000. Визначити, чи можна їх розмістити у відрізок арифметичної прогресії. Якщо потрібно, порядок чисел можна змінювати.
Відповідь дати у формі YES або NO.

	Приклад

	INPUT.TXT
	80 50 10 30 70 40 20 60 90

	OUTPUT.TXT
	YES


Розв’язок
Розмір статичної пам’яті (64 КБ) не дозволяє розмістити усі числа в одному масиві (на це необхідно до 400 КБ), отже задача потребує обробки даних "на місці", без їх збереження. Як же це зробити?
Перш за все, згадаймо властивості членів арифметичної прогресії:

· Різниця між сусідніми членами прогресії є постійною величиною (різниця прогресії d);
· Перший член прогресії (a0) є її найменшим членом, а всі інші члени виражаються через нього: ai=a0+i·d.
Якщо з даного набору чисел можна скласти відрізок арифметичної прогресії, то у неї є і перший член, і різниця прогресії, які ми можемо визначити: першим членом буде найменше із заданих чисел, а різницею – різниця між двома найменшими числами набору.

Таким чином, за один прохід по вхідних даних можна визначити a0 та d.
Проте необхідно перевірити, чи усі числа належать прогресії і чи слідує за кожним її членом наступний, ai+1=ai+d. Для цього при повторному проході по даних будемо відмічати, який член прогресії у нас вже один раз зустрівся. Для цієї мети найдоцільніше використати множини.
Заведемо масив множин s, у якому будемо зводити i-й біт j-ї множини в одиницю, коли чергове число ak=a0+((j-1)·256+(i-1))·d. Тоді, якщо вхідні дані складають арифметичну прогресію, то в нашому масиві множин рівно n послідовних біт будуть мати значення 1. Достатньо перевірити цю умову, щоб дати відповідь на поставлене в задачі запитання.
program Ariph_Progress;

const maxN = 400;

var fi,fo:text;

    i,n,num,a0,at,d,x:longint;

    Answer:string;

    S:array [0..maxN] of set of byte;

begin
  Assign (fi,'input.txt'); Reset (fi);

  a0:=MaxLongInt; at:=MaxLongInt; n:=0;

  while not SeekEOF(fi) do
     begin
       Read (fi,x);

       Inc (n);

       if x<a0 then a0:=x else
       if x<at then at:=x;

     end;

  Close (fi);

  d:=at-a0; Answer:='YES';

  Assign (fi,'input.txt'); Reset (fi);

  while (not SeekEOF(fi))and(Answer='YES') do
     begin
       Read (fi,x);

       if (x-a0) mod d=0 then
          begin
            num:=(x-a0) div d;

            if (num mod 256) in S[num div 256]

               then Answer:='NO'

               else Include (S[num div 256],num mod 256);

          end else Answer:='NO';

     end;

  Close (fi);

  Assign (fo,'output.txt'); Rewrite (fo);

  if Answer='YES' then
     begin
       for i:=0 to n-1 do
           if not((i mod 256) in S[i div 256]) then
              begin
                Answer:='NO';

                break;

              end;

     end;

  WriteLn (fo,Answer);

  Close (fo);

end.
При такому способі обробки даних ми виконуємо порядку 3·n операції, тобто за часом цілком вкладаємося в 1 секунду. І при цьому ми використовуємо не більше 13 КБ сегменту даних за рахунок вживання множин.
Підсумок
В цьому випуску ми розібрали два методи, за допомогою яких при розв’язку задачі можна обминути використання великих ресурсів. Першим з них є широко відомий метод динамічного програмування. Не буде перебільшенням, якщо сказати, що він використовується при розв’язанні 75% олімпіадних задач з програмування. Другий з розглянутих методів – використання масива множин – використовується не так часто, але стає незамінним помічником, коли необхідно обробити великий об’єм даних.

